


Calculating HCl Recycle Flow

$$Furnace_{Cl2.load} := [550, 600..800] \frac{m^3}{hr} = \begin{bmatrix} 550 \\ 600 \\ 650 \\ 700 \\ 750 \\ 800 \end{bmatrix} \frac{m^3}{hr}$$

$$\rho_{Cl2} := 3.15 \frac{kg}{m^3} \quad \rho_{H2O} := 1 \frac{tonne}{m^3}$$

$$HCl_{gas.prod} := \frac{36.5}{35.5} \cdot \rho_{Cl2} \cdot Furnace_{Cl2.load} = \begin{bmatrix} 1.7813 \\ 1.9432 \\ 2.1052 \\ 2.2671 \\ 2.429 \\ 2.591 \end{bmatrix} \frac{tonne}{hr}$$



$$HCl_{aq.3} := \frac{HCl_{gas.prod}}{0.31} = \begin{bmatrix} 5.7461 \\ 6.2685 \\ 6.7909 \\ 7.3133 \\ 7.8356 \\ 8.358 \end{bmatrix} \frac{tonne}{hr}$$

from DCS observations in comparison with Water.ideal.1

$$Leakage_{H2O.6} := 1.6 \frac{tonne}{hr}$$

Use eval when there is a line in the calculus of the variable

Use max or min in an index for solve, it found two solutions each time. Or use roots with an appropriate guess value for look only one.

$$Water_{ideal.1} := eval \left(\begin{array}{l} \text{for } n \in [1..length(HCl_{gas.prod})] \\ a_n := \max \left(\text{solve} \left(\frac{HCl_{gas.prod}_n}{HCl_{gas.prod}_n + x \frac{tonne}{hr}} = 0.31, x \right) \right) \\ a \frac{tonne}{hr} \end{array} \right) = \begin{bmatrix} 3.9648 \\ 4.3253 \\ 4.6857 \\ 5.0462 \\ 5.4066 \\ 5.767 \end{bmatrix} \frac{tonne}{hr}$$

w.t% diluted acid maintained in tank because of scrubber & recycle 4. This enters tail-gas absorber

$$DiluteAcid_{real.1} := [0, 0.02..0.14] = \begin{bmatrix} 0 \\ 2 \\ 4 \\ 6 \\ 8 \\ 10 \\ 12 \\ 14 \end{bmatrix} \%$$

Basic Scheme

$$Total_{H2O.Entering} := Leakage_{H2O.6} + x \frac{\text{tonne}}{\text{hr}} \cdot \left(1 - DiluteAcid_{real.1} \right) = \frac{16 + 9 \cdot x}{10} \frac{\text{tonne}}{\text{hr}}$$

$$Total_{HCl.gas.Entering} := HCl_{gas.prod} + x \frac{\text{tonne}}{\text{hr}} \cdot \left(DiluteAcid_{real.1} \right) = \frac{9198 + 355 \cdot x}{3550} \frac{\text{tonne}}{\text{hr}}$$

$$\text{solve} \left(\frac{Total_{HCl.gas.Entering}}{Total_{HCl.gas.Entering} + Total_{H2O.Entering}} = 0.31, x \right) \frac{\text{tonne}}{\text{hr}} = \left[\begin{array}{c} 4.5462 \\ 6.1513 \end{array} \right] \frac{\text{tonne}}{\text{hr}}$$

$$\text{max} \left(\text{solve} \left(\frac{Total_{HCl.gas.Entering}}{Total_{HCl.gas.Entering} + Total_{H2O.Entering}} = 0.31, x \right) \right) \frac{\text{tonne}}{\text{hr}} = 6.1513 \frac{\text{tonne}}{\text{hr}}$$

use eval

$$DiluteAcidFeed_{real.1} := \text{eval} \left(\begin{array}{l} \text{for } m \in [1..length(DiluteAcid_{real.1})] \\ \quad \Sigma H2O_{feed} := Leakage_{H2O.6} + x \frac{\text{tonne}}{\text{hr}} \cdot \left(1 - DiluteAcid_{real.1} \right) \\ \quad \text{for } n \in [1..length(HCl_{gas.prod})] \\ \quad \quad \Sigma HCl_{feed} := HCl_{gas.prod} + x \frac{\text{tonne}}{\text{hr}} \cdot DiluteAcid_{real.1} \\ \quad \quad a_{m,n} := \text{max} \left(\text{solve} \left(\frac{\Sigma HCl_{feed}}{\Sigma HCl_{feed} + \Sigma H2O_{feed}} = 0.31, x \right) \right) \\ \quad a \frac{\text{tonne}}{\text{hr}} \end{array} \right)$$

eval make the variable a global. So, you need to clear it like you do.

deNest Function

deNest("matrix") - Expand the nested matrix A.
This makes the composition of matrices by sub-matrices more readable.

```
deNest(A) := for i_Row ∈ [1..rows(A)]
  Clear(A_newRow)
  for i_Col ∈ [1..cols(A)]
    if IsDefined(A_newRow)
      A_newRow := augment(A_newRow, A_i_Row i_Col)
    else
      A_newRow := A_i_Row i_Col
  if IsDefined(A_new)
    A_new := stack(A_new, A_newRow)
  else
```

$A_{new} := A_{newRow}$
 A_{new}

deNest("matrix") - Expand the nested matrix A.
 This makes the composition of matrices by sub-matrices more readable.

This version is more efficient.

```
deNest(A) :=
  A_new := eval(matrix(0, cols(A_1)))
  for a ∈ A
    A_new := eval(stack(A_new, a))
  A_new
```


use eval


```
Table := eval(
  horizontal := stack("Diluted Feed% ↓ | Cl2 Load m³/h →", DiluteAcid_real.1 · 100)
  vertical := stack(Furnace_Cl2.load · 3600  $\frac{s}{m^3}$ , DiluteAcidFeed_real.1 ·  $\frac{3600 s}{1000 kg}$ )
  augment(horizontal, vertical)
)
```

"Diluted Feed% ↓ Cl2 Load m³/h →"	550	600	650	700	750	800
0	2.36	2.73	3.09	3.45	3.81	4.17
2	2.53	2.91	3.3	3.68	4.07	4.45
4	2.72	3.13	3.54	3.96	4.37	4.78
6	2.93	3.38	3.83	4.27	4.72	5.17
8	3.19	3.67	4.16	4.64	5.13	5.62
10	3.49	4.02	4.56	5.09	5.62	6.15
12	3.86	4.45	5.03	5.62	6.21	6.8
14	4.31	4.97	5.63	6.28	6.94	7.6

Plotting curves for various furnace Cl2 loading


```
for n ∈ [1..cols(DiluteAcidFeed_real.1)]
  y_n := col(DiluteAcidFeed_real.1 ·  $\frac{3600 s}{1000 kg}$ , n)
  plot_n := augment(DiluteAcid_real.1 · 100, y_n)
  label_n := augment(14, y_n_8, var2str(Furnace_Cl2.load_n ·  $3600 \frac{s}{m^3}$ ), 8)
```





$$\left\{ \text{augment} \left(\text{Furnace}_{\text{Cl2.load}} \cdot \frac{3600}{1} \frac{\text{s}}{\text{m}}, \text{Water}_{\text{ideal.1}} \cdot \frac{3600}{1000} \frac{\text{s}}{\text{kg}} \right) \right\}$$

Jacket Leakage 1.6m³/hr



{ plot
deNest (label)

Balance around production water tank

Clear (a) = 1


$$\text{DiluteAcid}_{\text{Feedvol.1}} := [3..7] \frac{\text{tonne}}{\text{hr}} = \begin{bmatrix} 3 \\ 4 \\ 5 \\ 6 \\ 7 \end{bmatrix} \frac{\text{tonne}}{\text{hr}}$$

$$\text{DiluteAcid}_{\text{FeedConc.1}} := [0.02, 0.04..0.10] = \begin{bmatrix} 2 \\ 4 \\ 6 \\ 8 \\ 10 \end{bmatrix} \%$$

$$\text{PureWater}_{\text{Feedvol.2}} := a \frac{\text{tonne}}{\text{hr}}$$

$$\text{Acid}_{\text{Feedvol.4}} := b \frac{\text{tonne}}{\text{hr}}$$

$$\text{Acid}_{\text{FeedConc.4}} := 0.31$$



Basic Scheme

$$\text{WaterBalance}_1 := \text{Acid}_{\text{Feedvol.4}} \cdot (1 - \text{Acid}_{\text{FeedConc.4}}) + \text{PureWater}_{\text{Feedvol.2}} - \text{DiluteAcid}_{\text{Feedvol.1}} \cdot (1 - \text{DiluteAcid}_{\text{FeedConc.1}})$$

$$\text{WaterBalance}_1 = \frac{69 \cdot b + 100 \cdot a - 540}{100} \frac{\text{tonne}}{\text{hr}}$$

$$\text{AcidBalance}_2 := \text{Acid}_{\text{Feedvol.4}} \cdot \text{Acid}_{\text{FeedConc.4}} - \text{DiluteAcid}_{\text{Feedvol.1}} \cdot \text{DiluteAcid}_{\text{FeedConc.1}}$$

$$\text{AcidBalance}_2 = \frac{-60 + 31 \cdot b}{100} \frac{\text{tonne}}{\text{hr}}$$

$$\begin{bmatrix} a \\ b \end{bmatrix} := \text{roots} \left(\begin{bmatrix} \text{WaterBalance}_1 \\ \text{AcidBalance}_2 \end{bmatrix}, \begin{bmatrix} a \\ b \end{bmatrix} \right) = \begin{bmatrix} 4.065 \\ 1.935 \end{bmatrix}$$

Clear (a, b) = 1

```
var := eval (
  for m ∈ [1..length(DiluteAcid_Feedvol.1)]
  for n ∈ [1..length(DiluteAcid_FeedConc.1)]
  WB_m_n := Acid_Feedvol.4 * (1 - Acid_FeedConc.4) + PureWater_Feedvol.2 - DiluteAcid_Feedvol.1_m * (1 - DiluteAcid_FeedConc.1_n)
  AB_m_n := Acid_Feedvol.4 * Acid_FeedConc.4 - DiluteAcid_Feedvol.1_m * DiluteAcid_FeedConc.1_n
  [j_m_n] := roots ( [WB_m_n], [a] )
  [k_m_n] := roots ( [AB_m_n], [b] )
  [j]
  [k]
)
```

$$\begin{bmatrix} \text{PureWater}_{\text{Feedvol.2}} \\ \text{Acid}_{\text{Feedvol.4}} \end{bmatrix} := \begin{bmatrix} \text{var}_1 \frac{\text{tonne}}{\text{hr}} \\ \text{var}_2 \frac{\text{tonne}}{\text{hr}} \end{bmatrix} = \begin{bmatrix} 2.806 & 2.613 & 2.419 & 2.226 & 2.032 \\ 3.742 & 3.484 & 3.226 & 2.968 & 2.71 \\ 4.677 & 4.355 & 4.032 & 3.71 & 3.387 \\ 5.613 & 5.226 & 4.839 & 4.452 & 4.065 \\ 6.548 & 6.097 & 5.645 & 5.194 & 4.742 \\ 0.194 & 0.387 & 0.581 & 0.774 & 0.968 \\ 0.258 & 0.516 & 0.774 & 1.032 & 1.29 \\ 0.323 & 0.645 & 0.968 & 1.29 & 1.613 \\ 0.387 & 0.774 & 1.161 & 1.548 & 1.935 \\ 0.452 & 0.903 & 1.355 & 1.806 & 2.258 \end{bmatrix} \frac{\text{tonne}}{\text{hr}}$$

$$\text{RecycleAcid}_{31\%} := \begin{cases} \text{horizontal} := \text{augment} \left(\text{"Feed flow tonne/hr} \downarrow \mid \text{Feed HCl\%} \rightarrow", \text{DiluteAcid}_{\text{FeedConc.1}}^T \cdot 100 \right) \\ \text{vertical} := \text{augment} \left(\text{DiluteAcid}_{\text{Feedvol.1}} \frac{\text{hr}}{\text{tonne}}, \text{Acid}_{\text{Feedvol.4}} \frac{\text{hr}}{\text{tonne}} \right) \\ \text{stack}(\text{horizontal}, \text{vertical}) \end{cases}$$

$$\text{PureWater}_{\text{Feed}} := \begin{cases} \text{horizontal} := \text{augment} \left(\text{"Feed flow tonne/hr} \downarrow \mid \text{Feed HCl\%} \rightarrow", \text{DiluteAcid}_{\text{FeedConc.1}}^T \cdot 100 \right) \\ \text{vertical} := \text{augment} \left(\text{DiluteAcid}_{\text{Feedvol.1}} \frac{\text{hr}}{\text{tonne}}, \text{PureWater}_{\text{Feedvol.2}} \frac{\text{hr}}{\text{tonne}} \right) \\ \text{stack}(\text{horizontal}, \text{vertical}) \end{cases}$$

$$\text{RecycleAcid}_{31\%} = \begin{bmatrix} \text{"Feed flow tonne/hr} \downarrow \mid \text{Feed HCl\%} \rightarrow" & 2 & 4 & 6 & 8 & 10 \\ 3 & 0.194 & 0.387 & 0.581 & 0.774 & 0.968 \\ 4 & 0.258 & 0.516 & 0.774 & 1.032 & 1.29 \\ 5 & 0.323 & 0.645 & 0.968 & 1.29 & 1.613 \\ 6 & 0.387 & 0.774 & 1.161 & 1.548 & 1.935 \\ 7 & 0.452 & 0.903 & 1.355 & 1.806 & 2.258 \end{bmatrix}$$

$$\text{PureWater}_{\text{Feed}} = \begin{bmatrix} \text{"Feed flow tonne/hr} \downarrow \mid \text{Feed HCl\%} \rightarrow" & 2 & 4 & 6 & 8 & 10 \\ 3 & 2.806 & 2.613 & 2.419 & 2.226 & 2.032 \\ 4 & 3.742 & 3.484 & 3.226 & 2.968 & 2.71 \\ 5 & 4.677 & 4.355 & 4.032 & 3.71 & 3.387 \\ 6 & 5.613 & 5.226 & 4.839 & 4.452 & 4.065 \\ 7 & 6.548 & 6.097 & 5.645 & 5.194 & 4.742 \end{bmatrix}$$